

# “Guessing game” assignment

Izaak van Dongen

October 30, 2017

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Boilerplate</b>	<b>2</b>
2.1	oracle . . . . .	2
2.2	read_lohi . . . . .	2
<b>3</b>	<b>Linear Search</b>	<b>3</b>
3.1	Positive integers . . . . .	3
3.2	All integers . . . . .	4
3.3	Rational numbers . . . . .	4
<b>4</b>	<b>Binary search</b>	<b>6</b>
4.1	Basic binary . . . . .	6
4.2	Bound-finding binary search . . . . .	7
4.3	Rational numbers . . . . .	8
<b>5</b>	<b>Interface</b>	<b>9</b>
<b>6</b>	<b>Source</b>	<b>9</b>

## Listings

1	oracle pseudocode . . . . .	2
2	oracle implementation . . . . .	2
3	read_lohi pseudocode . . . . .	3
4	read_lohi implementation . . . . .	3
5	Linear search on $\mathbb{N}$ pseudocode . . . . .	3
6	Linear search on $\mathbb{N}$ implementation . . . . .	4
7	Linear search on $\mathbb{Z}$ implementation . . . . .	4
8	Euclid’s algorithm in Pascal . . . . .	5
9	GCD behaviour for negative numbers . . . . .	6
10	Linear search on $\mathbb{Q}$ implementation in Pascal . . . . .	6
11	Basic binary search pseudocode . . . . .	6
12	Basic binary search in Pascal . . . . .	7
13	Bound-finding pseudocode . . . . .	7
14	Bound-finding subroutine in Pascal . . . . .	7
15	Binary search on $\mathbb{Q}$ in pseudocode . . . . .	8
16	Binary search on $\mathbb{Q}$ in Pascal . . . . .	8
17	Binary search on $\mathbb{Z}$ wrapper procedure . . . . .	9
18	Main interface loop . . . . .	9

## List of Figures

1	A Stern-Brocot tree . . . . .	8
---	-------------------------------	---

## Introduction

This is an assignment about guessing games and appropriate strategies. This is linked to searching algorithms, although in this case it’s a sort of searching algorithm where you’re searching for the item, rather than searching for the index of a known item.

Both binary searches and linear searches can be adapted to that end, as both work by comparisons of the target and “guesses”. These are analogous to asking the user questions - in the case of linear search, the question “is your number equal to ...?”, and in the case of binary search, “is your number greater than ...?”

Therefore, essentially we need only to implement binary and linear search where the “comparison function” delegates to the user. However, interpolation search, which has the complexity  $O(\log(\log(n)))$  (on uniform data sets, which we can assume a user-generated target is), requires knowledge of the *value* of the target, so it can be interpolated on the current upper and lower bounds. This means we will only achieve the  $O(\log(n))$  complexity of binary search.

Sadly, we also won’t be able to implement any such search algorithm for the set of reals  $\mathbb{R}$ , or any set with equivalent or higher cardinality (eg  $\mathbb{C} = \mathbb{R}^2 \Rightarrow |\mathbb{C}| = |\mathbb{R}|$ ). This is because they are not a countably infinite set - there exists no bijection between them and the natural numbers. If we assume we could write a search algorithm to guess a real number, each of the algorithm’s guesses could be labeled with a natural number, implying there does exist some enumeration of the reals. This contradicts lots of established knowledge (eg Cantor’s diagonal argument) and so we know this won’t be possible.

Code for this assignment has been written in Pascal, targeted at and tested in fpc (version 3.0.0+dfsg-2), using mode `{ $MODE OBJFPC }`. This document was developed in L<sup>A</sup>T<sub>E</sub>X.

Where I’ve deemed it necessary, I’ve left comments, although often the code really is relatively simple, and the functions are sufficiently described by their names and the simplicity of the operations (“self-documented”).

At some points, I use some algebraic notation. I use  $\wedge$ , which means logical and, to indicate that multiple statements are true.

My general model/definition for a “search algorithm” is an algorithm that can ask only boolean questions of a user, and must use ideally as few of these as possible to guess a target number that the user knows.

Generally, I have provided pseudocode for my algorithms. My pseudocode is probably pretty close to structured english, but I feel like if I were to make it much more code-like it would more or less be Pascal/Python.

## Boilerplate

To create a program that performs such a search to guess a number, some utility functions will be needed, that aren’t really particularly algorithmic. I’m implementing these in a Pascal unit, named `PUser.pas`. Having these in a separate unit means the code concerned with the actual algorithms can be more “purely” expressed. It also means that that code can become more modular itself, as another caller could feasibly implement their own boilerplate, and the abstracted algorithms will still work.

### oracle

This is a function that can retrieve a boolean value from the user. It’s named “oracle” as it acts as the comparison oracle the search functions must appeal to. The only parameter it takes is a message to display. The unix-inspired ruling it uses is that if an input starts with a “y”, it is considered affirmative, and negative otherwise. The broad approach it uses is like this:

```
1 If the user's response is empty, the result is negative
2 If the user's response is not empty and begins with a lower or uppercase "y", the result is positive
3 Otherwise, the result is negative
```

Listing 1: oracle pseudocode

The catch for lower or uppercase is implemented by coercing the entire string to lowercase, using `LowerCase`. It’s implemented as:

```
1 function oracle(msg: string): boolean;
2 var
3     usr_input: string;
4 begin
5     write(msg);
6     readln(usr_input);
7     if (length(usr_input) > 0) and (LowerCase(usr_input)[1] = 'y') then
8         oracle := true
9     else
10        oracle := false;
11 end;
```

Listing 2: oracle implementation

NB: The lack of semicolon in line 8 is on purpose, as this is how an if-else statement works in Pascal (as it’s considered one statement, the whole affair needs one semicolon).

### read\_lohi

NB this function is currently unused in the source, but it was used for testing one of the earlier adaptations of the binary search algorithm. It also provides some interesting talking points about call signatures in Pascal.

I wanted to provide some facility for setting the upper bound and lower bound of a binary search. As this function wants to kind of “return” two things, it has a nontrivial call signature, which is worth thinking about. One possible solution might be to define some kind of **Bounds** container class that can hold both values.

However, Pascal provides a way to output multiple values, and this is through an **out** parameter. It could also be done using a **var** parameter, although there’s a slight difference. Both of these work, and in both cases the variable is passed by reference, but **out** is slightly more specific - it’s used when the input value of the variable is unneeded, which in this case is true, as I’m only concerned with using it as a channel for output. Once I’ve declared something as an **out** parameter, I can assign to it and the caller can then use the new value (like the **readln** function).

I will be reading these from the command line arguments, as I’m developing it as a console application. As this is quite a simple application, I will just manually parse arguments with some if statements. The function will roughly do the following:

```

1 Gather default values for the upper bound and lower bound
2 If a first command line argument is present
3     Set the lower bound to this number
4 Otherwise, set it to the default
5 If a second command line argument is present
6     Set the upper bound to this number
7 Otherwise, set it to the default
8 If either argument given was not an integer
9     Set both to the defaults

```

Listing 3: read\_lohi pseudocode

This is then implemented in Pascal using **ParamStr** and **ParamCount**. Interestingly, as it doesn’t return anything in the conventional manner, this isn’t a function but a procedure.

```

1 procedure read_lohi(lo_default, hi_default: integer;
2                   out low_val, hi_val: integer);
3 begin
4     try
5         if ParamCount >= 1 then
6             low_val := StrToInt(ParamStr(1))
7         else
8             low_val := lo_default;
9         if ParamCount >= 2 then
10            hi_val := StrToInt(ParamStr(2))
11        else
12            hi_val := hi_default;
13    except
14        on E: EConvertError do begin
15            writeln('Conversion error occurred, reverting to defaults');
16            low_val := lo_default;
17            hi_val := hi_default;
18        end;
19    end;
20 end;

```

Listing 4: read\_lohi implementation

## Linear Search

One approach to this is by a “linear search”. This involved, basically, a kind of “brute force” approach - sequentially making guesses until one is correct. The single advantage of this algorithm is that it has no upper bound, and can even feasibly be made to work without a lower bound.

### Positive integers

The most simple approach is to assume the number is some  $x$  such that

$$x \in \{0\} \cup \mathbb{N} \quad (1)$$

ie the set of natural numbers including 0. Guesses can then be made sequentially like so:

```

1 Set the current ‘guess’ to 0
2 While the guess is wrong
3     Increment the guess by one

```

Listing 5: Linear search on  $\mathbb{N}$  pseudocode

In Pascal, this could be implemented like so:

```

1 function linear_search : integer ;
2 var
3     i : integer = 0 ;
4 begin
5     while not oracle('Is your number equal to ' + IntToStr(i) + '? ') do
6         i := i + 1 ;
7         linear_search := i ;
8 end ;

```

Listing 6: Linear search on  $\mathbb{N}$  implementation

## All integers

The flaw in the previous program is that if the user sneakily decides to think of a negative number, this program won't ever terminate. This can be solved by using some enumeration of the set of all integers  $\mathbb{Z}$ . The simplest of these is the sequence 0, 1, -1, 2, -2, 3, -3, 4, -4, 5, -5, 6, -6 ... This sequence doesn't duplicate 0, and has the property of "spreading out" from 0. A sequence kind of like this with a lot of overhead could be represented by simply generating the set of natural numbers  $\mathbb{N}$  and then taking each of these and its negation. However, if we consider it as more of a mathematical sequence, it *can* be represented by a closed formula

$$U_n = (-1)^n \left\lfloor \frac{n}{2} \right\rfloor \quad (2)$$

However, it's much better represented inductively with a condition:

$$U_1 = 0$$

$$U_{n+1} = \begin{cases} -U_n + 1 & \text{if } x \leq 0 \\ -U_n & \text{if } x > 0 \end{cases}$$

This can also quite easily be implemented in code:

```

1 function linear_search : integer ;
2 var
3     i : integer = 0 ;
4 begin
5     while not oracle('Is your number equal to ' + IntToStr(i) + '? ') do
6         if i <= 0 then
7             i := -i + 1
8         else
9             i := -i ;
10        linear_search := i ;
11 end ;

```

Listing 7: Linear search on  $\mathbb{Z}$  implementation

This does not need any double loops, duplicate code to explicitly negate each number, or extra boilerplate logic to prevent duplicating a 0.

## Rational numbers

Interestingly, a linear search can also be generalised to work on the set of rational numbers  $\mathbb{Q}$ . This is closely related to the fact that the cardinality of the set of rationals

$$|\mathbb{Q}| = |\mathbb{N}| \quad (3)$$

This means there exists a bijection from  $\mathbb{N}$  to  $\mathbb{Q}$ , and hence there must exist some enumeration of the elements of  $\mathbb{Q}$ , which can be obtained by applying the bijection to all the naturals in order.

We can easily come up with such an enumeration by considering the set of rationals as a grid with two integer axes representing numerator and denominator (in other words, the set  $\mathbb{Z}^2$ ). We can then "spiral outwards" from the origin. This will be easiest to do in squares aligned with the axes, through which we move in the anticlockwise direction, although it could also be done in the diagonal direction. The first part of the grid, in terms of co-ordinate pairs, will be like this:

$$\begin{aligned}
&(-2, 2)(-1, 2)(0, 2)(1, 2)(2, 2) \\
&(-2, 1)(-1, 1)(0, 1)(1, 1)(2, 1) \\
&(-2, 0)(-1, 0)(0, 0)(1, 0)(2, 0) \\
&(-2, -1)(-1, -1)(0, -1)(1, -1)(2, -1) \\
&(-2, -2)(-1, -2)(0, -2)(1, -2)(2, -2)
\end{aligned}$$

Where the first couple of terms of the enumeration are:

$$\begin{aligned}
&(0,0), \\
&(1,1),(0,1),(-1,1),(-1,0),(-1,-1),(0,-1),(1,-1),(1,0), \\
&(2,1),(2,2),(1,2),(0,2),(-1,2),(-2,2),(-2,1),(-2,0),... \\
&(3,1),(3,2),...
\end{aligned} \tag{4}$$

Note that the first term is  $\frac{0}{0}$ . This technically is not a rational number as its value is undefined, but we can deal with that later - first we can establish the enumeration, and then “filter” it for unwanted pairings. Note also that this enumeration will not preserve any conception of size of fractions -  $\frac{1}{2}$  will occur long before  $\frac{1}{100}$ .

We know that this path will reach all of the points on this grid, hence this enumeration will eventually reach the target - and this is where a linear search is useful. First, we will need to consider how we can implement this enumeration. We can consider the desired behaviour on each side of a square, and the conditions describing each such side.

Let us consider two sequences  $X$  and  $Y$ , where we say that

$$\frac{X_i}{Y_i} \tag{5}$$

is the  $i$ th element of our enumeration.

We know the first term is  $\frac{0}{0}$ . Hence, we have  $X_1=0 \wedge Y_1=0$

We also say that we jump to the next layer of square when we “hit” the positive x-axis. To start the next square, we move to the right one and up one, as if we just move to the right, we stay on the axis and move right again. This means  $X_i \geq 0 \wedge Y_i = 0 \Rightarrow X_{i+1} = X_i + 1 \wedge Y_{i+1} = 1$

If we are on the right side of the square, we know that  $X_i$  is positive. We also know that  $-X_i \leq Y_i < X_i$ . When this is true, we want to move “up”, so increment  $Y$ . We also know that  $Y_i = 0$  is considered a special case. Hence, we say that

$$X_i > 0 \wedge -X_i \leq Y_i < X_i \wedge Y_i \neq 0 \Rightarrow X_{i+1} = X_i \wedge Y_{i+1} = Y_i + 1 \tag{6}$$

We can produce similar definitions for the other three sides:

$$\begin{aligned}
Y_i > 0 \wedge -Y_i < X_i \leq Y_i &\Rightarrow X_{i+1} = X_i - 1 \wedge Y_{i+1} = Y_i \\
X_i < 0 \wedge X_i < Y_i \leq -X_i &\Rightarrow X_{i+1} = X_i \quad \wedge Y_{i+1} = Y_i - 1 \\
Y_i < 0 \wedge Y_i \leq X_i < -Y_i &\Rightarrow X_{i+1} = X_i + 1 \wedge Y_{i+1} = Y_i
\end{aligned}$$

Now that we have a way to produce such an enumeration, we can consider a filtering. The first condition is that we skip any number with a denominator of 0, as this is undefined. The second condition is that we should only consider fractions in their simplest form - ie the numerator and denominator are coprime. This means we don't duplicate any rationals and so obtain a strict bijection between the naturals and the rationals. Knowing that

$$A \text{ and } B \text{ are coprime} \Leftrightarrow \gcd(A, B) = 1 \tag{7}$$

We can simply calculate the gcd of  $X$  and  $Y$  using Euclid's algorithm, implemented like so in Pascal:

```

1 function gcd(a, b: integer): integer;
2 var
3   temp: integer;
4 begin
5   while b <> 0 do begin
6     temp := b;
7     b := a mod b;
8     a := temp;

```

```

9   end;
10  gcd := a;
11  end;

```

Listing 8: Euclid’s algorithm in Pascal

This also works how we want for negative numbers - printing a couple of examples:

```

1 gcd 1 2 is 1
2 gcd 1 -2 is 1
3 gcd -1 2 is -1
4 gcd -1 -2 is -1

```

Listing 9: GCD behaviour for negative numbers

We can see that only one of the pairs 1,-2 and -1,2 are considered coprime, which is correct as  $\frac{1}{-2} = \frac{-1}{2} = -\frac{1}{2}$ . We can also see that -1 and -2 are not considered coprime, which is again correct as it is equivalent to  $\frac{1}{2}$ .

We can now put all of this together to obtain a fully featured linear search algorithm on the set of rational numbers:

```

1 procedure linear_search(out o_x: integer;
2                        out o_y: integer);
3 var
4     x: integer = 0;
5     y: integer = 0;
6 begin
7     while (y = 0) or (gcd(x, y) <> 1) or
8           (not oracle('Is your number equal to ' + IntToStr(x) + '/' + IntToStr(y) + '? ')) do
9         if (x >= 0) and (y = 0) then begin
10             x := x + 1;
11             y := y + 1;
12         end else if (x > 0) and (-x <= y) and (y < x) then
13             y := y + 1;
14         else if (y > 0) and (-y < x) and (x <= y) then
15             x := x - 1;
16         else if (x < 0) and (x < y) and (y <= -x) then
17             y := y - 1;
18         else
19             x := x + 1;
20     o_x := x;
21     o_y := y;
22 end;

```

Listing 10: Linear search on  $\mathbb{Q}$  implementation in Pascal

This again uses the out parameter trick to return multiple values. It is also a little terser than the fully qualified mathematical conditions given earlier, as some conditions are mutually exclusive/eliminated. Interestingly, the body of the while loop is technically a single statement, so no begin ...end block is needed. The condition of the loop is that it keeps going if the number is one to be skipped, or the user does not indicate that the guess is correct.

Another thing this code makes use of is short-circuit boolean evaluation. All of the “filter” checks have been compounded into one boolean expression, using the or operator. Even though there is a call to oracle in this statement, the call will only actually be evaluated if neither of the filter checks flag up - if it doesn’t pass the filter, it will short-circuit so the call to oracle will be skipped.

A parting consideration is that the linear search is very “secure”, in a sense - a malicious user will not be able to confuse the linear search in any way, they will only be able to jeopardise their own experience - any sequence of answers for a linear search will seem sensible to it, even if they are intended maliciously. Really, a user’s only power is to lie about what their number was, which a simple command line program won’t realistically be able to defend against.

## Binary search

At last! The binary search algorithm is significantly faster than the linear search, in the average case (being  $O(\log(n))$ ).

### Basic binary

In its most basic form, the binary search might work something like this in pseudocode:

```

1 Establish a lower bound and a strict upper bound
2 While the difference between the bounds is greater than one
3     Guess a number in the middle of these bounds
4     If the guess was too high, move the upper bound down to the guess
5     If the guess was too low, move the lower bound up to the guess

```

Listing 11: Basic binary search pseudocode

Note that this algorithm will rely on some "magic" upper and lower bound, for now. I've used some slightly careful consideration of range arithmetic and definitions - by having a strict upper bound, when the length of the range is one, the only remaining candidate in the range is the lower bound. This is one easy way to prevent off-by-one errors and too much case-checking here. It can be implemented in Pascal quite trivially, like so:

```

1 function binary_search(lower, upper: integer): integer;
2 var
3     mid: integer;
4     is_ge: boolean;
5 begin
6     while upper - lower > 1 do begin
7         mid := (upper + lower) div 2;
8         is_ge := oracle('Is your number greater than or equal to ' + IntToStr(mid) + '? ');
9         if is_ge then
10             lower := mid
11         else
12             upper := mid;
13     end;
14     binary_search := lower;
15 end;

```

Listing 12: Basic binary search in Pascal

This function takes two parameters, the bounds, and returns the guessed number. It is also relatively immune to malicious users - again, every sequence of answers leads to a correct guess, assuming all the answers were correct.

### Bound-finding binary search

Our binary search could be made significantly less naïve by not putting a bounds restriction on the user. We can actually determine a bound quite quickly, assuming nothing other than that the user's number will be an integer.

Continuing the theme of binary and powers of two, we can simply check the "bound" between each of the adjacent powers of 2, ie first 1-2, then 2-4, then 4-8 (note that these are the good old fashioned strict upper bound ranges, so there is no overlap). This will converge to some number  $n$  in logarithmic time, again. However, this does not account for negative numbers - what we can easily do is simply consider the *absolute* value of the user's number  $|n|$ . We need then only ask if it is negative once, at the end. If it is negative, we will need to "flip" the range, bearing in mind to increase the new lower bound and decrease the new upper bound, to make them non-strict and strict, respectively. This leaves the single edge case where the user's number is 0, as this is not an integer power of 2. We can explicitly check for this to get around it (not very elegant, but simple enough).

In pseudocode, this might be implemented as follows:

```

1 If the user's number is 0, return the range 0-1
2 For each pair a and b of powers of 2
3     If a <= |n| < b
4         Remember this range, and stop guessing ranges
5 If the user's number is positive, return this range
6 Otherwise, return the 'flipped' range

```

Listing 13: Bound-finding pseudocode

This then comes to the following implementation in Pascal:

```

1 procedure binary_find_bounds(out o_lower: integer;
2                             out o_upper: integer);
3 var
4     lower: integer = 1;
5     upper: integer = 2;
6 begin
7     writeln('This is the stage where I determine some bounds on your number. I '
8           , ' will be asking questions about the *magnitude* of your number, '
9           , ' so watch out. ');
10    if oracle('Is your number 0? ') then begin
11        o_lower := 0;
12        o_upper := 1;
13    end else begin
14        while not oracle('Is your number some x such that ' +
15                        IntToStr(lower) + ' <= |x| < ' + IntToStr(upper) + '? ') do begin
16            lower := lower * 2;
17            upper := upper * 2;
18        end;
19        if oracle('Is your number greater than or equal to 0? ') then begin
20            o_lower := lower;
21            o_upper := upper;
22        end else begin

```

```

23         o_lower := - upper + 1;
24         o_upper := - lower - 1;
25     end;
26 end;
27 end;

```

Listing 14: Bound-finding subroutine in Pascal

Using this, we can successfully perform a binary search on the set of all integers  $\mathbb{Z}$ . In fact, we only need to first call this subroutine to determine the bounds, and then let the previous function from listing 12 do the rest.

## Rational numbers

Interestingly, theoretically we already have enough “material” to put together a binary search on the set of all rationals. We can use the previously defined enumeration, and then simply ask the user to think of what the index of their number is in that enumeration, and then perform a binary search as normal to find that integer. Here, the concept of “greater than” has been entirely abstracted away from the comparison of actual rationals.

However, this makes some rather theoretical demands of the user, and is better left as a perverse though experiment. There are far better ways to approach this. Namely, there exists a binary search tree of the set of rationals - the “Stern-Brocot” tree, which uses fractional mediants:




Figure 1: A Stern-Brocot tree

This also addresses the concern about comparison - this tree does preserve a proper ordering of the rationals, due to the fact that the mediant of two fractions lies strictly in between those fractions.

Interestingly, this tree has kind of similar behaviour to what we encountered in the bound-finding function - it only terminates on positive nonzero numbers. We can get around this in the same way - first check for 0, ask questions about the absolute value of the user’s number and then ask if it is negative:

```

1 Ask if the user’s number is 0
2 Set the lower and upper bound to 0/1 and 1/0
3 While you haven’t guessed the user’s number
4     Generate the mediant of the bounds
5     If this is the number, stop
6     If this is too high, move the upper bound down to the mediant
7     Otherwise, move the lower bound up to the mediant

```

Listing 15: Binary search on  $\mathbb{Q}$  in pseudocode

This can be implemented like so in Pascal:

```

1 procedure binary_search(out o_x: integer;
2                       out o_y: integer);
3 var
4     mid_x, mid_y: integer;
5     lo_x: integer = 0;
6     lo_y: integer = 1;
7     hi_x: integer = 1;
8     hi_y: integer = 0;
9     is_ge: boolean;
10 begin
11     if oracle('Is your number equal to 0? ') then begin

```



```

12     o_x := 0;
13     o_y := 1;
14 end else begin
15     repeat
16         mid_x := lo_x + hi_x;
17         mid_y := lo_y + hi_y;
18         is_ge := oracle('is your number q such that '
19                         + IntToStr(mid_x) + '/' + IntToStr(mid_y)
20                         + ' <= |q|? ');
21         if is_ge then begin
22             lo_x := mid_x;
23             lo_y := mid_y;
24         end else begin
25             hi_x := mid_x;
26             hi_y := mid_y;
27         end;
28     until is_ge and oracle('is your number q such that |q| = '
29                           + IntToStr(mid_x) + '/' + IntToStr(mid_y) + '? ');
30     o_x := mid_x;
31     o_y := mid_y;
32     if oracle('Is your number less than 0? ') then
33         o_x := -mid_x;
34 end;
35 end;

```

Listing 16: Binary search on  $\mathbb{Q}$  in Pascal

## Interface

Having implemented all of the algorithms, it was now time to write an interface. First, I wrote a number of procedures to go with each algorithm, prefixed with “perform.”. They mainly just printed some information and formatted the results to the user. For example, to go with the binary search on  $\mathbb{Z}$ :

```

1 procedure perform_bsearch;
2 var
3     guess: integer;
4     lower, upper: integer;
5 begin
6     writeln('welcome to the binary search game! Think of an integer. ');
7     binary_find_bounds(lower, upper);
8     guess := binary_search(lower, upper);
9     writeln('your number was ', guess);
10 end;

```

Listing 17: Binary search on  $\mathbb{Z}$  wrapper procedure

Using such procedures, I then wrote the following short program:

```

1 while true do begin
2     if oracle('Would you like to perform an integer binary search? ') then
3         perform_bsearch
4     else if oracle('Would you like to perform a rational binary search? ') then
5         perform_fracbsearch
6     else if oracle('Would you like to perform an integer linear search? ') then
7         perform_lsearch
8     else if oracle('Would you like to perform a rational linear search? ') then
9         perform_fraclsearch;
10 end;

```

Listing 18: Main interface loop

## Source

The full project in its directory structure, including this document, can be found at [https://github.com/elterminad0r/assignment\\_guessing](https://github.com/elterminad0r/assignment_guessing).