

“Guessing game” assignment

Izaak van Dongen

October 30, 2017

Contents

1	Introduction	1
2	Boilerplate	1
2.1	oracle	2
2.2	read_lohi	2
3	Linear Search	3
3.1	Positive integers	3
3.2	All integers	3
3.3	Rational numbers	4
4	Source	6

Listings

1	oracle pseudocode	2
2	oracle implementation	2
3	read_lohi pseudocode	2
4	read_lohi implementation	2
5	Linear search on \mathbb{N} pseudocode	3
6	Linear search on \mathbb{N} implementation	3
7	Linear search on \mathbb{Z} implementation	3
8	Euclid’s algorithm in Pascal	5
9	GCD behaviour for negative numbers	5
10	Linear search on \mathbb{Q} implementation in Pascal	5

Introduction

This is an assignment about guessing games and appropriate strategies. This is linked to searching algorithms, although in this case it’s a sort of searching algorithm where you’re searching for the item, rather than searching for the index of a known item.

Both binary searches and linear searches can be adapted to that end, as both work by comparisons of the target and “guesses”. These are analogous to asking the user questions - in the case of linear search, the question “is your number equal to ...?”, and in the case of binary search, “is your number greater than ...?”

Therefore, essentially we need only to implement binary and linear search where the “comparison function” delegates to the user. However, interpolation search, which has the complexity $O(\log(\log(n)))$ (on uniform data sets, which we can assume a user-generated target is), requires knowledge of the *value* of the target, so it can be interpolated on the current upper and lower bounds. This means we will only achieve the $O(\log(n))$ complexity of binary search.

Code for this assignment has been written in Pascal, targeted at and tested in fpc (version 3.0.0+dfsg-2), using mode `{ $MODE OBJFPC }`. This document was developed in L^AT_EX.

Where I’ve deemed it necessary, I’ve left comments, although often the code really is relatively simple, and the functions are sufficiently described by their names and the simplicity of the operations (“self-documented”).

At some points, I use some algebraic notation. I use \wedge , which means logical and, to indicate that multiple statements are true.

Boilerplate

To create a program that performs such a search to guess a number, some utility functions will be needed, that aren’t really particularly algorithmic. I’m implementing these in a Pascal unit, named `PUser.pas`. Having these in a separate unit means the code concerned with the actual algorithms can be more “purely” expressed. It also means that that code can become more modular itself, as another caller could feasibly implement their own boilerplate, and the abstracted algorithms will still work.

oracle

This is a function that can retrieve a boolean value from the user. It's named "oracle" as it acts as the comparison oracle the search functions must appeal to. The only parameter it takes is a message to display. The unix-inspired ruling it uses is that if an input starts with a "y", it is considered affirmative, and negative otherwise. The broad approach it uses is like this:

```
1 If the user's response is empty, the result is negative
2 If the user's response is not empty and begins with a lower or uppercase "y", the result is positive
3 Otherwise, the result is negative
```

Listing 1: oracle pseudocode

The catch for lower or uppercase is implemented by coercing the entire string to lowercase, using `LowerCase`. It's implemented as:

```
1 function oracle(msg: string): boolean;
2 var
3     usr_input: string;
4 begin
5     write(msg);
6     readln(usr_input);
7     if (length(usr_input) > 0) and (LowerCase(usr_input)[1] = 'y') then
8         oracle := true
9     else
10        oracle := false;
11 end;
```

Listing 2: oracle implementation

NB: The lack of semicolon in line 8 is on purpose, as this is how an if-else statement works in Pascal (as it's considered one statement, the whole affair needs one semicolon).

read_lohi

I wanted to provide some facility for setting the upper bound and lower bound of a binary search. As this function wants to kind of "return" two things, it has a nontrivial call signature, which is worth thinking about. One possible solution might be to define some kind of `Bounds` container class that can hold both values.

However, Pascal provides a way to output multiple values, and this is through an `out` parameter. It could also be done using a `var` parameter, although there's a slight difference. Both of these work, and in both cases the variable is passed by reference, but `out` is slightly more specific - it's used when the input value of the variable is unneeded, which in this case is true, as I'm only concerned with using it as a channel for output. Once I've declared something as an `out` parameter, I can assign to it and the caller can then use the new value (like the `readln` function).

I will be reading these from the command line arguments, as I'm developing it as a console application. As this is quite a simple application, I will just manually parse arguments with some if statements. The function will roughly do the following:

```
1 Gather default values for the upper bound and lower bound
2 If a first command line argument is present
3     Set the lower bound to this number
4 Otherwise, set it to the default
5 If a second command line argument is present
6     Set the upper bound to this number
7 Otherwise, set it to the default
8 If either argument given was not an integer
9     Set both to the defaults
```

Listing 3: read_lohi pseudocode

This is then implemented in Pascal using `ParamStr` and `ParamCount`. Interestingly, as it doesn't return anything in the conventional manner, this isn't a function but a procedure.

```
1 procedure read_lohi(lo_default, hi_default: integer;
2                     out low_val, hi_val: integer);
3 begin
4     try
5         if ParamCount >= 1 then
6             low_val := StrToInt(ParamStr(1))
7         else
8             low_val := lo_default;
9         if ParamCount >= 2 then
10            hi_val := StrToInt(ParamStr(2))
11        else
```

```

12         hi_val := hi_default;
13     except
14         on E: EConvertError do begin
15             writeln('Conversion error occurred, reverting to defaults');
16             low_val := lo_default;
17             hi_val := hi_default;
18         end;
19     end;
20 end;

```

Listing 4: read_lohi implementation

Linear Search

One approach to this is by a “linear search”. This involved, basically, a kind of “brute force” approach - sequentially making guesses until one is correct. The single advantage of this algorithm is that it has no upper bound, and can even feasibly be made to work without a lower bound.

Positive integers

The most simple approach is to assume the number is some x such that

$$x \in \{0\} \cup \mathbb{N} \quad (1)$$

ie the set of natural numbers including 0. Guesses can then be made sequentially like so:

```

1 Set the current ‘guess’ to 0
2 While the guess is wrong
3     Increment the guess by one

```

Listing 5: Linear search on \mathbb{N} pseudocode

In Pascal, this could be implemented like so:

```

1 function linear_search: integer;
2 var
3     i: integer = 0;
4 begin
5     while not oracle('Is your number equal to ' + IntToStr(i) + '? ') do
6         i := i + 1;
7         linear_search := i;
8 end;

```

Listing 6: Linear search on \mathbb{N} implementation

All integers

The flaw in the previous program is that if the user sneakily decides to think of a negative number, this program won’t ever terminate. This can be solved by using some enumeration of the set of all integers \mathbb{Z} . The simplest of these is the sequence 0, 1, -1, 2, -2, 3, -3, 4, -4, 5, -5, 6, -6 ... This sequence doesn’t duplicate 0, and has the property of “spreading out” from 0. A sequence kind of like this with a lot of overhead could be represented by simply generating the set of natural numbers \mathbb{N} and then taking each of these and its negation. However, if we consider it as more of a mathematical sequence, it *can* be represented by a closed formula

$$U_n = (-1)^n \left\lfloor \frac{n}{2} \right\rfloor \quad (2)$$

However, it’s much better represented inductively with a condition:

$$U_1 = 0$$

$$U_{n+1} = \begin{cases} -U_n + 1 & \text{if } x \leq 0 \\ -U_n & \text{if } x > 0 \end{cases}$$

This can also quite easily be implemented in code:

```

1 function linear_search: integer;
2 var
3     i: integer = 0;
4 begin
5     while not oracle('Is your number equal to ' + IntToStr(i) + '? ') do
6         if i <= 0 then

```

```

7         i := -i + 1
8     else
9         i := -i;
10    linear_search := i;
11 end;

```

Listing 7: Linear search on \mathbb{Z} implementation

This does not need any double loops, duplicate code to explicitly negate each number, or extra boilerplate logic to prevent duplicating a 0.

Rational numbers

Interestingly, a linear search can also be generalised to work on the set of rational numbers \mathbb{Q} . This is closely related to the fact that the cardinality of the set of rationals

$$|\mathbb{Q}| = |\mathbb{N}| \quad (3)$$

This means there exists a bijection from \mathbb{N} to \mathbb{Q} , and hence there must exist some enumeration of the elements of \mathbb{Q} , which can be obtained by applying the bijection to all the naturals in order.

We can easily come up with such an enumeration by considering the set of rationals as a grid with two integer axes representing numerator and denominator (in other words, the set \mathbb{Z}^2). We can then “spiral outwards” from the origin. This will be easiest to do in squares aligned with the axes, through which we move in the anticlockwise direction, although it could also be done in the diagonal direction. The first part of the grid, in terms of co-ordinate pairs, will be like this:

$$\begin{aligned}
 &(-2, 2)(-1, 2)(0, 2)(1, 2)(2, 2) \\
 &(-2, 1)(-1, 1)(0, 1)(1, 1)(2, 1) \\
 &(-2, 0)(-1, 0)(0, 0)(1, 0)(2, 0) \\
 &(-2, -1)(-1, -1)(0, -1)(1, -1)(2, -1) \\
 &(-2, -2)(-1, -2)(0, -2)(1, -2)(2, -2)
 \end{aligned}$$

Where the first couple of terms of the enumeration are:

$$\begin{aligned}
 &(0,0), \\
 &(1,1),(0,1),(-1,1),(-1,0),(-1,-1),(0,-1),(1,-1),(1,0), \\
 &(2,1),(2,2),(1,2),(0,2),(-1,2),(-2,2),(-2,1),(-2,0),\dots \\
 &(3,1),(3,2),\dots
 \end{aligned} \quad (4)$$

Note that the first term is $\frac{0}{0}$. This technically is not a rational number as its value is undefined, but we can deal with that later - first we can establish the enumeration, and then “filter” it for unwanted pairings. Note also that this enumeration will not preserve any conception of size of fractions - $\frac{1}{2}$ will occur long before $\frac{1}{100}$.

We know that this path will reach all of the points on this grid, hence this enumeration will eventually reach the target - and this is where a linear search is useful. First, we will need to consider how we can implement this enumeration. We can consider the desired behaviour on each side of a square, and the conditions describing each such side.

Let us consider two sequences X and Y , where we say that

$$\frac{X_i}{Y_i} \quad (5)$$

is the i th element of our enumeration.

We know the first term is $\frac{0}{0}$. Hence, we have $X_1 = 0 \wedge Y_1 = 0$

We also say that we jump to the next layer of square when we “hit” the positive x-axis. To start the next square, we move to the right one and up one, as if we just move to the right, we stay on the axis and move right again. This means $X_i \geq 0 \wedge Y_i = 0 \Rightarrow X_{i+1} = X_i + 1 \wedge Y_{i+1} = 1$

If we are on the right side of the square, we know that X_i is positive. We also know that $-X_i \leq Y_i < X_i$. When this is true, we want to move “up”, so increment Y . We also know that $Y_i = 0$ is considered a special case. Hence, we say that

$$X_i > 0 \wedge -X_i \leq Y_i < X_i \wedge Y_i \neq 0 \Rightarrow X_{i+1} = X_i \wedge Y_{i+1} = Y_i + 1 \quad (6)$$

We can produce similar definitions for the other three sides:

$$\begin{aligned} Y_i > 0 \wedge -Y_i < X_i \leq Y_i &\Rightarrow X_{i+1} = X_i - 1 \wedge Y_{i+1} = Y_i \\ X_i < 0 \wedge X_i < Y_i \leq -X_i &\Rightarrow X_{i+1} = X_i \quad \wedge Y_{i+1} = Y_i - 1 \\ Y_i < 0 \wedge Y_i \leq X_i < -Y_i &\Rightarrow X_{i+1} = X_i + 1 \wedge Y_{i+1} = Y_i \end{aligned}$$

Now that we have a way to produce such an enumeration, we can consider a filtering. The first condition is that we skip any number with a denominator of 0, as this is undefined. The second condition is that we should only consider fractions in their simplest form - ie the numerator and denominator are coprime. This means we don't duplicate any rationals and so obtain a strict bijection between the naturals and the rationals. Knowing that

$$A \text{ and } B \text{ are coprime} \Leftrightarrow \gcd(A, B) = 1 \quad (7)$$

We can simply calculate the gcd of X and Y using Euclid's algorithm, implemented like so in Pascal:

```

1 function gcd(a, b: integer): integer;
2 var
3   temp: integer;
4 begin
5   while b <> 0 do begin
6     temp := b;
7     b := a mod b;
8     a := temp;
9   end;
10  gcd := a;
11 end;
```

Listing 8: Euclid's algorithm in Pascal

This also works how we want for negative numbers - printing a couple of examples:

```

1 gcd 1 2 is 1
2 gcd 1 -2 is 1
3 gcd -1 2 is -1
4 gcd -1 -2 is -1
```

Listing 9: GCD behaviour for negative numbers

We can see that only one of the pairs 1,-2 and -1,2 are considered coprime, which is correct as $\frac{1}{-2} = \frac{-1}{2} = -\frac{1}{2}$. We can also see that -1 and -2 are not considered coprime, which is again correct as it is equivalent to $\frac{1}{2}$.

We can now put all of this together to obtain a fully featured linear search algorithm on the set of rational numbers:

```

1 procedure linear_search(out o_x: integer;
2   out o_y: integer);
3 var
4   x: integer = 0;
5   y: integer = 0;
6 begin
7   while (y = 0) or (gcd(x, y) <> 1) or
8     (not oracle('Is your number equal to ' + IntToStr(x) + '/' + IntToStr(y) + '? ')) do
9     if (x >= 0) and (y = 0) then begin
10      x := x + 1;
11      y := y + 1;
12    end else if (x > 0) and (-x <= y) and (y < x) then
13      y := y + 1;
14    else if (y > 0) and (-y < x) and (x <= y) then
15      x := x - 1;
16    else if (x < 0) and (x < y) and (y <= -x) then
17      y := y - 1;
18    else
19      x := x + 1;
20  o_x := x;
21  o_y := y;
22 end;
```

Listing 10: Linear search on \mathbb{Q} implementation in Pascal

This again uses the `out` parameter trick to return multiple values. It is also a little terser than the fully qualified mathematical conditions given earlier, as some conditions are mutually exclusive/eliminated. Interestingly, the body of the while loop is technically a single statement, so no `begin ...end` block is needed. The condition of the loop is that it keeps going if the number is one to be skipped, or the user does not indicate that the guess is correct.

Source

The full project in its directory structure, including this document, can be found at https://github.com/elterminad0r/assignment_guessing.